HW04Fall2016

In[3011]:=
```
Clear["Global`*"]
```

In[3012]:=
```
(* Problem 1 Part (a) *)
```

In[3013]:=
```
rcm =
 ell * {Sin[theta1[t]], Cos[theta1[t]]} + (b / 2) * {Sin[theta2[t]], Cos[theta2[t]]}
```

Out[3013]= $\left\{\text{ell Sin[theta1[t]]} + \frac{1}{2}\text{ b Sin[theta2[t]], ell Cos[theta1[t]]} + \frac{1}{2}\text{ b Cos[theta2[t]]}\right\}$

In[3014]:=
```
vcm = D[rcm, t]
```

Out[3014]= $\left\{\text{ell Cos[theta1[t]] theta1}'\text{[t]} + \frac{1}{2}\text{ b Cos[theta2[t]] theta2}'\text{[t]}, -\text{ell Sin[theta1[t]] theta1}'\text{[t]} - \frac{1}{2}\text{ b Sin[theta2[t]] theta2}'\text{[t]}\right\}$

In[3015]:=
```
Tcm = Simplify[ (1 / 2) * M * vcm.vcm]
```

Out[3015]= $\frac{1}{8}\text{ M }\left(4\text{ ell}^2\text{ theta1}'\text{[t]}^2 + 4\text{ b ell Cos[theta1[t] - theta2[t]] theta1}'\text{[t] theta2}'\text{[t]} + \text{b}^2\text{ theta2}'\text{[t]}^2\right)$

In[3016]:=
```
Trot = (1 / 2) * (M * b^2 / 12) * theta2'[t]^2
```

Out[3016]= $\frac{1}{24}\text{ b}^2\text{ M theta2}'\text{[t]}^2$

In[3017]:=
```
V = M * g * rcm.{0, -1}
```

Out[3017]= $\text{g M }\left(-\text{ell Cos[theta1[t]]} - \frac{1}{2}\text{ b Cos[theta2[t]]}\right)$

In[3018]:=
```
L = Expand[ (Tcm + Trot) - V]
```

Out[3018]= $\text{ell g M Cos[theta1[t]]} + \frac{1}{2}\text{ b g M Cos[theta2[t]]} + \frac{1}{2}\text{ ell}^2\text{ M theta1}'\text{[t]}^2 + \frac{1}{2}\text{ b ell M Cos[theta1[t] - theta2[t]] theta1}'\text{[t] theta2}'\text{[t]} + \frac{1}{6}\text{ b}^2\text{ M theta2}'\text{[t]}^2$

In[3019]:=
```
(* Make small-angle (quadratic) approximation by hand: *)
```

In[3020]:=
```
Series[Series[V, {theta1[t], 0, 2}], {theta2[t], 0, 2}]
```

Out[3020]= $\left(\left(-\frac{1}{2}\text{ b g M} - \text{ell g M}\right) + \frac{1}{4}\text{ b g M theta2[t]}^2 + \text{O[theta2[t]]}^3\right) + \frac{1}{2}\text{ ell g M theta1[t]}^2 + \text{O[theta1[t]]}^3$

In[3021]:=
```
tmp = Normal[%];
Vsmall = tmp - (tmp /. {theta1[t] → 0, theta2[t] → 0})
```

Out[3022]= $\frac{1}{2}$ ell g M theta1[t]^2 + $\frac{1}{4}$ b g M theta2[t]^2

In[3023]:=
```
T = Tcm + Trot
```

Out[3023]= $\frac{1}{24}$ b^2 M theta2′[t]^2 + $\frac{1}{8}$ M (4 ell^2 theta1′[t]^2 + 4 b ell Cos[theta1[t] - theta2[t]] theta1′[t] theta2′[t] + b^2 theta2′[t]^2)

In[3024]:=
```
Tsmall = Expand[T /. {theta1[t] - theta2[t] → 0}]
```

Out[3024]= $\frac{1}{2}$ ell^2 M theta1′[t]^2 + $\frac{1}{2}$ b ell M theta1′[t] theta2′[t] + $\frac{1}{6}$ b^2 M theta2′[t]^2

In[3025]:=
```
Lsmall = Tsmall - Vsmall
```

Out[3025]= $-\frac{1}{2}$ ell g M theta1[t]^2 $-$ $\frac{1}{4}$ b g M theta2[t]^2 + $\frac{1}{2}$ ell^2 M theta1′[t]^2 + $\frac{1}{2}$ b ell M theta1′[t] theta2′[t] + $\frac{1}{6}$ b^2 M theta2′[t]^2

In[3026]:=
```
(* More elegant way to make the small-oscillation (quadratic) approximation: *)
```

In[3027]:=
```
tmp1 = L /. {theta1[t] → s * theta1[t], theta2[t] → s * theta2[t],
   theta1'[t] → s * theta1'[t], theta2'[t] → s * theta2'[t]}
```

Out[3027]= ell g M Cos[s theta1[t]] + $\frac{1}{2}$ b g M Cos[s theta2[t]] + $\frac{1}{2}$ ell^2 M s^2 theta1′[t]^2 + $\frac{1}{2}$ b ell M s^2 Cos[s theta1[t] - s theta2[t]] theta1′[t] theta2′[t] + $\frac{1}{6}$ b^2 M s^2 theta2′[t]^2

In[3028]:=
```
tmp2 = Series[tmp1, {s, 0, 2}]
```

Out[3028]= $\left(\frac{\text{b g M}}{2}\right.$ + ell g M$\Big)$ + $\frac{1}{12}$ (−6 ell g M theta1[t]^2 − 3 b g M theta2[t]^2 + 6 ell^2 M theta1′[t]^2 + 6 b ell M theta1′[t] theta2′[t] + 2 b^2 M theta2′[t]^2) s^2 + O[s]^3

In[3029]:=
```
tmp3 = Simplify[Normal[tmp2]]
```

Out[3029]= $\frac{1}{12}$ M (6 b g + 12 ell g + s^2 (−6 ell g theta1[t]^2 − 3 b g theta2[t]^2 + 6 ell^2 theta1′[t]^2 + 6 b ell theta1′[t] theta2′[t] + 2 b^2 theta2′[t]^2))

In[3030]:=
```
LsmallElegant = Expand[(tmp3 /. s → 1) - (tmp3 /. s → 0)]
```

Out[3030]= $-\frac{1}{2}$ ell g M theta1[t]^2 $-$ $\frac{1}{4}$ b g M theta2[t]^2 + $\frac{1}{2}$ ell^2 M theta1′[t]^2 + $\frac{1}{2}$ b ell M theta1′[t] theta2′[t] + $\frac{1}{6}$ b^2 M theta2′[t]^2

In[3031]:=
```
Lsmall - LsmallElegant
```

Out[3031]= 0

In[3032]:=
```
(* Get small-
 oscillations equations of motion and find the normal mode frequencies *)
zero1 = D[D[Lsmall, theta1'[t]], t] - D[Lsmall, theta1[t]]
zero2 = D[D[Lsmall, theta2'[t]], t] - D[Lsmall, theta2[t]]
```

Out[3032]= $\text{ell}\ g\ M\ \text{theta1}[t] + \text{ell}^2\ M\ \text{theta1}''[t] + \frac{1}{2}\ b\ \text{ell}\ M\ \text{theta2}''[t]$

Out[3033]= $\frac{1}{2}\ b\ g\ M\ \text{theta2}[t] + \frac{1}{2}\ b\ \text{ell}\ M\ \text{theta1}''[t] + \frac{1}{3}\ b^2\ M\ \text{theta2}''[t]$

In[3034]:=
```
zzero1 = zero1 /.
  {theta1[t] → A, theta1''[t] → -omsq * A, theta2[t] → B, theta2''[t] → -omsq * B}
zzero2 = zero2 /. {theta1[t] → A, theta1''[t] → -omsq * A,
   theta2[t] → B, theta2''[t] → -omsq * B}
```

Out[3034]= $A\ \text{ell}\ g\ M - \frac{1}{2}\ b\ B\ \text{ell}\ M\ \text{omsq} - A\ \text{ell}^2\ M\ \text{omsq}$

Out[3035]= $\frac{1}{2}\ b\ B\ g\ M - \frac{1}{3}\ b^2\ B\ M\ \text{omsq} - \frac{1}{2}\ A\ b\ \text{ell}\ M\ \text{omsq}$

In[3036]:=
```
sol1 = Solve[zzero1 == 0, B]
sol2 = Solve[zzero2 == 0, B]
```

Out[3036]= $\left\{\left\{B \to \frac{2\ (A\ g - A\ \text{ell}\ \text{omsq})}{b\ \text{omsq}}\right\}\right\}$

Out[3037]= $\left\{\left\{B \to -\frac{3\ A\ \text{ell}\ \text{omsq}}{-3\ g + 2\ b\ \text{omsq}}\right\}\right\}$

In[3038]:=
```
solve = Solve[(B /. sol1[[1]]) == (B /. sol2[[1]]), omsq]
```

Out[3038]= $\left\{\left\{\text{omsq} \to -\left(\left(-4\ b\ g - 6\ \text{ell}\ g + \sqrt{\left(-24\ b\ \text{ell}\ g^2 + (4\ b\ g + 6\ \text{ell}\ g)^2\right)}\right) \Big/ (2\ b\ \text{ell})\right)\right\},\right.$
$\left.\left\{\text{omsq} \to \left(4\ b\ g + 6\ \text{ell}\ g + \sqrt{\left(-24\ b\ \text{ell}\ g^2 + (4\ b\ g + 6\ \text{ell}\ g)^2\right)}\right) \Big/ (2\ b\ \text{ell})\right\}\right\}$

In[3039]:=
```
omega1 = Simplify[Sqrt[omsq /. solve[[1]]], Assumptions → {b > 0, ell > 0, g > 0}]
omega2 = Simplify[Sqrt[omsq /. solve[[2]]], Assumptions → {b > 0, ell > 0, g > 0}]
```

Out[3039]= $\sqrt{\left(\frac{1}{b\ \text{ell}}\left(2\ b + 3\ \text{ell} - \sqrt{4\ b^2 + 6\ b\ \text{ell} + 9\ \text{ell}^2}\right)\ g\right)}$

Out[3040]= $\sqrt{\left(\frac{1}{b\ \text{ell}}\left(2\ b + 3\ \text{ell} + \sqrt{4\ b^2 + 6\ b\ \text{ell} + 9\ \text{ell}^2}\right)\ g\right)}$

In[3041]:=
```
{omega1, omega2} /. {g → 1, ell → 1, b → 1 / 4}
```

Out[3041]= $\left\{2\ \sqrt{\frac{7}{2} - \frac{\sqrt{43}}{2}},\ 2\ \sqrt{\frac{7}{2} + \frac{\sqrt{43}}{2}}\right\}$

In[3042]:=
```
ratio = FullSimplify[omega2 / omega1, Assumptions → {g > 0, ell > 0, b > 0}]
```

Out[3042]= $\sqrt{\left(\left(2\ b + 3\ \text{ell} + \sqrt{4\ b^2 + 6\ b\ \text{ell} + 9\ \text{ell}^2}\right) \Big/ \left(2\ b + 3\ \text{ell} - \sqrt{4\ b^2 + 6\ b\ \text{ell} + 9\ \text{ell}^2}\right)\right)}$

In[3043]:=
```
(* This ratio can be simplified better as follows: *)
```

In[3044]:=
```
P = 2*b + 3*ell;
Q = Sqrt[4*b^2 + 6*b*ell + 9*ell^2]
```

Out[3045]= $\sqrt{4\,b^2 + 6\,b\,ell + 9\,ell^2}$

In[3046]:=
```
rsimple = Simplify[(P + Q) / Sqrt[P^2 - Q^2], Assumptions → {b > 0, ell > 0}]
```

Out[3046]= $\left(2\,b + 3\,ell + \sqrt{4\,b^2 + 6\,b\,ell + 9\,ell^2}\right) \Big/ \left(\sqrt{6}\,\sqrt{b\,ell}\right)$

In[3047]:=
```
FullSimplify[ratio / rsimple, Assumptions → {b > 0, ell > 0}]
```

Out[3047]= 1

In[3048]:=
```
(* Set up the small-oscillations problem using our standard formalism *)
```

In[3049]:=
```
Tsmall
```

Out[3049]= $\frac{1}{2}\,ell^2\,M\,theta1'[t]^2 + \frac{1}{2}\,b\,ell\,M\,theta1'[t]\,theta2'[t] + \frac{1}{6}\,b^2\,M\,theta2'[t]^2$

In[3050]:=
```
tmat = M*{{ell^2, b*ell/2}, {b*ell/2, b^2/3}};
MatrixForm[tmat]
```

Out[3051]//MatrixForm=

$$\begin{pmatrix} ell^2\,M & \frac{b\,ell\,M}{2} \\ \frac{b\,ell\,M}{2} & \frac{b^2\,M}{3} \end{pmatrix}$$

In[3052]:=
```
Vsmall
```

Out[3052]= $\frac{1}{2}\,ell\,g\,M\,theta1[t]^2 + \frac{1}{4}\,b\,g\,M\,theta2[t]^2$

In[3053]:=
```
vmat = {{ell*g*M, 0}, {0, b*g*M / 2}};
MatrixForm[vmat]
```

Out[3054]//MatrixForm=

$$\begin{pmatrix} ell\,g\,M & 0 \\ 0 & \frac{b\,g\,M}{2} \end{pmatrix}$$

In[3055]:=
```
(* Check that tmat and vmat matrices agree with Lsmall: *)
```

In[3056]:=
```
ttmp = Expand[(1 / 2) * {theta1'[t], theta2'[t]}.tmat.{theta1'[t], theta2'[t]}]
```

Out[3056]= $\frac{1}{2}\,ell^2\,M\,theta1'[t]^2 + \frac{1}{2}\,b\,ell\,M\,theta1'[t]\,theta2'[t] + \frac{1}{6}\,b^2\,M\,theta2'[t]^2$

In[3057]:=
```
vtmp = Expand[(1 / 2) * {theta1[t], theta2[t]}.vmat.{theta1[t], theta2[t]}]
```

Out[3057]= $\frac{1}{2}\,ell\,g\,M\,theta1[t]^2 + \frac{1}{4}\,b\,g\,M\,theta2[t]^2$

In[3058]:=
```
Simplify[Lsmall - (ttmp - vtmp)]
```

Out[3058]= 0

In[3059]:=
```
(* Get the normal mode frequencies by the standard formalism
 and show that they agree with direct Lagrangian result: *)
```

In[3060]:=
```
Det[-omsq * tmat + vmat]
```

Out[3060]= $\frac{1}{2}\text{b ell g}^2\text{M}^2-\frac{1}{3}\text{b}^2\text{ell g M}^2\text{omsq}-\frac{1}{2}\text{b ell}^2\text{g M}^2\text{omsq}+\frac{1}{12}\text{b}^2\text{ell}^2\text{M}^2\text{omsq}^2$

In[3061]:=
```
sol = Solve[% == 0, omsq]
```

Out[3061]= $\{\{\text{omsq}\to\left(4\text{ b g}+6\text{ ell g}-\sqrt{\left(-24\text{ b ell g}^2+(-4\text{ b g}-6\text{ ell g})^2\right)}\right)/(2\text{ b ell})\},$
$\{\text{omsq}\to\left(4\text{ b g}+6\text{ ell g}+\sqrt{\left(-24\text{ b ell g}^2+(-4\text{ b g}-6\text{ ell g})^2\right)}\right)/(2\text{ b ell})\}\}$

In[3062]:=
```
om1 = FullSimplify[Sqrt[omsq /. sol[[1]]], Assumptions → {g > 0, ell > 0, b > 0}]
om2 = FullSimplify[Sqrt[omsq /. sol[[2]]], Assumptions → {g > 0, ell > 0, b > 0}]
```

Out[3062]= $\sqrt{\left(\frac{1}{\text{b ell}}\left(2\text{ b}+3\text{ ell}-\sqrt{4\text{ b}^2+6\text{ b ell}+9\text{ ell}^2}\right)\text{g}\right)}$

Out[3063]= $\sqrt{\left(\frac{1}{\text{b ell}}\left(2\text{ b}+3\text{ ell}+\sqrt{4\text{ b}^2+6\text{ b ell}+9\text{ ell}^2}\right)\text{g}\right)}$

In[3064]:=
```
om1 - omega1
om2 - omega2
```

Out[3064]= 0

Out[3065]= 0

In[3066]:=
```
(* For curiosity, see how omega2/omega1 varies with b/ell: *)
```

In[3067]:=
```
Plot[ratio /. ell → 1, {b, .2, 10}]
```

Out[3067]=



In[3068]:=
```
(* Could make the motion periodic if make ratio=
 omega2/omega1 an integer (or a ratio of two integers) *)
Solve[(ratio /. ell → 1) == 4, b]
N[%]
```

Out[3068]= $\left\{\left\{b \to \frac{3}{256}\left(161 - 17\sqrt{33}\right)\right\}, \left\{b \to \frac{3}{256}\left(161 + 17\sqrt{33}\right)\right\}\right\}$

Out[3069]= {{b → 0.742294}, {b → 3.03114}}

In[3070]:=
```
Solve[(ratio /. ell → 1) == 5, b]
N[%]
```

Out[3070]= $\left\{\left\{b \to \frac{3}{100}\left(119 - 13\sqrt{69}\right)\right\}, \left\{b \to \frac{3}{100}\left(119 + 13\sqrt{69}\right)\right\}\right\}$

Out[3071]= {{b → 0.330417}, {b → 6.80958}}

In[3072]:=
```
FindMinimum[(ratio /. ell → 1), {b, 1}]
```

Out[3072]= {3.73205, {b → 1.5}}

In[3073]:=
```
ratio /. {ell → 1, b → 0.75}
```

Out[3073]= 3.99215

In[3074]:=
```
(* Find the normal mode amplitudes for b = ell/4 *)
```

In[3075]:=
```
matrix = -omsq * tmat + vmat;
MatrixForm[matrix]
```

Out[3076]//MatrixForm=

$$\begin{pmatrix} \text{ell g M} - \text{ell}^2 \text{ M omsq} & -\frac{1}{2}\text{ b ell M omsq} \\ -\frac{1}{2}\text{ b ell M omsq} & \frac{\text{b g M}}{2} - \frac{1}{3}\text{b}^2\text{ M omsq} \end{pmatrix}$$

In[3077]:=
```
matrix1 = Simplify[(matrix /. omsq → omega1^2) /. {ell → 1, g → 1, M → 1, b → 1/4}];
matrix1 // MatrixForm
```

Out[3078]//MatrixForm=

$$\begin{pmatrix} -13 + 2\sqrt{43} & \frac{1}{4}\left(-7 + \sqrt{43}\right) \\ \frac{1}{4}\left(-7 + \sqrt{43}\right) & \frac{1}{24}\left(-4 + \sqrt{43}\right) \end{pmatrix}$$

In[3079]:=
```
Det[matrix1]
```

Out[3079]= 0

In[3080]:=
```
tmatrix = tmat /. {ell → 1, g → 1, M → 1, b → 1/4}
```

Out[3080]= $\left\{\left\{1, \frac{1}{8}\right\}, \left\{\frac{1}{8}, \frac{1}{48}\right\}\right\}$

In[3081]:=
```
A1 = {A11, A12};
Solve[{matrix1.A1 == 0, A1.tmatrix.A1 == 1}, {A11, A12}]
```

Out[3082]=
$$\left\{\left\{A11 \to -\sqrt{\frac{2\left(59-8\sqrt{43}\right)}{43-4\sqrt{43}}},\ A12 \to \left(6\left(-7\sqrt{\frac{2\left(59-8\sqrt{43}\right)}{43-4\sqrt{43}}}+\sqrt{\frac{86\left(59-8\sqrt{43}\right)}{43-4\sqrt{43}}}\right)\right)\Big/\left(-4+\sqrt{43}\right)\right\},\ \left\{A11 \to \sqrt{\frac{2\left(59-8\sqrt{43}\right)}{43-4\sqrt{43}}},\ A12 \to \left(6\left(7\sqrt{\frac{2\left(59-8\sqrt{43}\right)}{43-4\sqrt{43}}}-\sqrt{\frac{86\left(59-8\sqrt{43}\right)}{43-4\sqrt{43}}}\right)\right)\Big/\left(-4+\sqrt{43}\right)\right\}\right\}$$

In[3083]:=
```
FullSimplify[%]
```

Out[3083]=
$$\left\{\left\{A11 \to -\sqrt{2-\frac{8}{\sqrt{43}}},\ A12 \to -4\sqrt{6-\frac{39}{\sqrt{43}}}\right\},\ \left\{A11 \to \sqrt{2-\frac{8}{\sqrt{43}}},\ A12 \to 4\sqrt{6-\frac{39}{\sqrt{43}}}\right\}\right\}$$

In[3084]:=
```
A1vec = FullSimplify[A1 /. %[[2]]]
```

Out[3084]=
$$\left\{\sqrt{2-\frac{8}{\sqrt{43}}},\ 4\sqrt{6-\frac{39}{\sqrt{43}}}\right\}$$

In[3085]:=
```
matrix2 = Simplify[(matrix /. omsq → omega2^2) /. {ell → 1, g → 1, M → 1, b → 1/4}];
matrix2 // MatrixForm
```

Out[3086]//MatrixForm=
$$\begin{pmatrix} -13-2\sqrt{43} & \frac{1}{4}\left(-7-\sqrt{43}\right) \\ \frac{1}{4}\left(-7-\sqrt{43}\right) & \frac{1}{24}\left(-4-\sqrt{43}\right) \end{pmatrix}$$

In[3087]:=
```
A2 = {A21, A22};
```

In[3088]:=
```
FullSimplify[Solve[{matrix2.A2 == 0, A2.tmatrix.A2 == 1}, {A21, A22}]]
```

Out[3088]=
$$\left\{\left\{A21 \to -\sqrt{2+\frac{8}{\sqrt{43}}},\ A22 \to 4\sqrt{6+\frac{39}{\sqrt{43}}}\right\},\ \left\{A21 \to \sqrt{2+\frac{8}{\sqrt{43}}},\ A22 \to -4\sqrt{6+\frac{39}{\sqrt{43}}}\right\}\right\}$$

In[3089]:=
```
A2vec = FullSimplify[A2 /. %[[2]]]
```

Out[3089]=
$$\left\{\sqrt{2+\frac{8}{\sqrt{43}}},\ -4\sqrt{6+\frac{39}{\sqrt{43}}}\right\}$$

In[3090]:=
```
FullSimplify[A1vec.tmatrix.A1vec]
```
Out[3090]= 1

In[3091]:=
```
FullSimplify[A1vec.tmatrix.A2vec]
```
Out[3091]= 0

In[3092]:=
```
thvec = FullSimplify[
  (A1vec * c1 * Cos[omega1 * t] + A2vec * c2 * Cos[omega2 * t]) /. {g → 1, ell → 1, b → 1 / 4}]
```
Out[3092]=
$$\left\{\sqrt{2-\frac{8}{\sqrt{43}}}\ c1\ \mathrm{Cos}\left[\sqrt{14-2\sqrt{43}}\ t\right]+\sqrt{2+\frac{8}{\sqrt{43}}}\ c2\ \mathrm{Cos}\left[\sqrt{2\left(7+\sqrt{43}\right)}\ t\right],\ 4\sqrt{6-\frac{39}{\sqrt{43}}}\ c1\ \mathrm{Cos}\left[\sqrt{14-2\sqrt{43}}\ t\right]-4\sqrt{6+\frac{39}{\sqrt{43}}}\ c2\ \mathrm{Cos}\left[\sqrt{2\left(7+\sqrt{43}\right)}\ t\right]\right\}$$

In[3093]:=
```
tmp = FullSimplify[Solve[(thvec /. t → 0) == {0, theta20}, {c1, c2}]]
```
Out[3093]=
$$\left\{\left\{c1\to\sqrt{\frac{1}{96}+\frac{1}{24\sqrt{43}}}\ theta20,\ c2\to-\frac{1}{4}\sqrt{\frac{1}{6}-\frac{2}{3\sqrt{43}}}\ theta20\right\}\right\}$$

In[3094]:=
```
thetavec = FullSimplify[thvec /. tmp[[1]]]
```
Out[3094]=
$$\left\{\frac{1}{4\sqrt{43}}3\ theta20\left(\mathrm{Cos}\left[\sqrt{14-2\sqrt{43}}\ t\right]-\mathrm{Cos}\left[\sqrt{2\left(7+\sqrt{43}\right)}\ t\right]\right),\ \frac{1}{2\sqrt{43}}\ theta20\left(\left(-5+\sqrt{43}\right)\mathrm{Cos}\left[\sqrt{14-2\sqrt{43}}\ t\right]+\left(5+\sqrt{43}\right)\mathrm{Cos}\left[\sqrt{2\left(7+\sqrt{43}\right)}\ t\right]\right)\right\}$$

In[3095]:=
```
theta1 = Simplify[thetavec[[1]] /. theta20 → 2 * Pi / 180]
```
Out[3095]=
$$\left(\pi\left(\mathrm{Cos}\left[\sqrt{14-2\sqrt{43}}\ t\right]-\mathrm{Cos}\left[\sqrt{2\left(7+\sqrt{43}\right)}\ t\right]\right)\right)\Big/\left(120\sqrt{43}\right)$$

In[3096]:=
```
Plot[theta1, {t, 0, 10}]
```
Out[3096]=


In[3097]:=
```
theta2 = Simplify[thetavec[[2]] /. theta20 → 2 * Pi / 180]
```

Out[3097]=
$$\left(\pi\left(\left(-5+\sqrt{43}\right)\text{Cos}\left[\sqrt{14-2\sqrt{43}}\ t\right]+\left(5+\sqrt{43}\right)\text{Cos}\left[\sqrt{2\left(7+\sqrt{43}\right)}\ t\right]\right)\right)\Big/\left(180\sqrt{43}\right)$$

In[3098]:=
```
Plot[theta2, {t, 0, 10}]
```

Out[3098]=


In[3099]:=
```
(* Motion of center of stick  -- it's quite close to sinusoidal,
though of course not actually periodic *)
```

In[3100]:=
```
Plot[theta1 + (1 / 8) * theta2, {t, 0, 10}]
```

Out[3100]=


In[3101]:=
```
(* Let Mathematica find the motion directly by solving the small-
oscillation Eqs: *)
```

In[3102]:=
```
Clear[theta1, theta2]
```

In[3103]:=
```
Lsmall
```

Out[3103]=
$$-\frac{1}{2}\text{ell g M theta1}[t]^2-\frac{1}{4}\text{b g M theta2}[t]^2+\frac{1}{2}\text{ell}^2\text{ M theta1}'[t]^2+\frac{1}{2}\text{b ell M theta1}'[t]\text{ theta2}'[t]+\frac{1}{6}\text{b}^2\text{ M theta2}'[t]^2$$

In[3104]:=
```
p1small = D[Lsmall, theta1'[t]]
p2small = D[Lsmall, theta2'[t]]
```

Out[3104]= $\text{ell}^2\, M\, \text{theta1}'[t] + \frac{1}{2}\, b\, \text{ell}\, M\, \text{theta2}'[t]$

Out[3105]= $\frac{1}{2}\, b\, \text{ell}\, M\, \text{theta1}'[t] + \frac{1}{3}\, b^2\, M\, \text{theta2}'[t]$

In[3106]:=
```
f1small = D[Lsmall, theta1[t]]
f2small = D[Lsmall, theta2[t]]
```

Out[3106]= $-\text{ell}\, g\, M\, \text{theta1}[t]$

Out[3107]= $-\frac{1}{2}\, b\, g\, M\, \text{theta2}[t]$

In[3108]:=
```
dsolve = DSolve[{D[p1small, t] == f1small, D[p2small, t] == f2small,
    theta1[0] == 0, theta2[0] == 2 * Pi / 180, theta1'[0] == 0, theta2'[0] == 0} /.
   {g → 1, ell → 1, b → 1 / 4}, {theta1[t], theta2[t]}, t]
```

Out[3108]= $\left\{\left\{\text{theta1}[t] \to \frac{1}{120}\, \pi\, \text{RootSum}\left[24 + 28\, \#1^2 + \#1^4\, \&,\ \frac{e^{t\, \#1}}{14 + \#1^2}\, \&\right],\right.\right.$
$\left.\left.\text{theta2}[t] \to \frac{1}{360}\, \pi\, \text{RootSum}\left[24 + 28\, \#1^2 + \#1^4\, \&,\ \frac{4\, e^{t\, \#1} + e^{t\, \#1}\, \#1^2}{14 + \#1^2}\, \&\right]\right\}\right\}$

In[3109]:=
```
Plot[theta1[t] /. dsolve[[1]], {t, 0, 10}]
```

Out[3109]=



In[3110]:=
```
Plot[(theta1[t] + (1 / 8) * theta2[t]) /. dsolve[[1]], {t, 0, 10}]
```

Out[3110]=



In[3111]:=
```
(* Now find the motion by numerically
 integrating the exact equations of motion *)
```

In[3112]:=
```
p1 = Simplify[D[L, theta1'[t]]]
f1 = Simplify[D[L, theta1[t]]]
```

Out[3112]= $\frac{1}{2}$ ell M (2 ell theta1′[t] + b Cos[theta1[t] - theta2[t]] theta2′[t])

Out[3113]= $-\frac{1}{2}$ ell M (2 g Sin[theta1[t]] + b Sin[theta1[t] - theta2[t]] theta1′[t] theta2′[t])

In[3114]:=
```
p2 = Simplify[D[L, theta2'[t]]]
f2 = Simplify[D[L, theta2[t]]]
```

Out[3114]= $\frac{1}{6}$ b M (3 ell Cos[theta1[t] - theta2[t]] theta1′[t] + 2 b theta2′[t])

Out[3115]= $-\frac{1}{2}$ b M (g Sin[theta2[t]] - ell Sin[theta1[t] - theta2[t]] theta1′[t] theta2′[t])

In[3116]:=
```
nd = NDSolve[{D[p1, t] == f1, D[p2, t] == f2, theta1[0] == 0,
    theta2[0] == theta20, theta1'[0] == 0, theta2'[0] == 0} /. {ell → 1, g → 1,
    M → 1, b → 1 / 4, theta20 → 2 * Pi / 180}, {theta1[t], theta2[t]}, {t, 0, 20}]
```

Out[3116]= {{theta1[t] → InterpolatingFunction[Domain: {{0., 20.}} Output: scalar][t],

theta2[t] → InterpolatingFunction[Domain: {{0., 20.}} Output: scalar][t]}}

In[3117]:=
```
Plot[theta1[t] /. nd[[1]], {t, 0, 10}]
```

Out[3117]=



In[3118]:=
```
(* Numerical solution with larger starting angle,
which was carefully chosen to make the solution approximately periodic *)
```

In[3123]:=
```
Clear[theta20]
nd2 = NDSolve[
  {D[p1, t] == f1, D[p2, t] == f2, theta1[0] == 0, theta2[0] == theta20, theta1'[0] == 0,
    theta2'[0] == 0} /. {ell → 1, g → 1, M → 1, b → 1 / 4, theta20 → 129.82 * Pi / 180},
  {theta1[t], theta2[t]}, {t, 0, 1000}]
```

Out[3124]=
```
{{theta1[t] → InterpolatingFunction[ Domain: {{0., 1.00×10^3}} Output: scalar ][t],
  theta2[t] → InterpolatingFunction[ Domain: {{0., 1.00×10^3}} Output: scalar ][t]}}
```

In[3125]:=
```
Plot[theta1[t] /. nd2[[1]], {t, 0, 20}]
```

Out[3125]=




```
(* Not really periodic -- different at much later time *)
```

In[3129]:=
```
Plot[theta1[t] /. nd2[[1]], {t, 380, 400}]
```

Out[3129]=